
Description of Project

The following design and analysis was conducted on a combined cycle Brayton / Rankine power plant.  A typical schematic is shown in Figure 1.  A table of the characteristics and parameters that were used and/or assumed is presented in Table 1.  While the assignment provided the option to consider reheat in the Rankine cycle, we chose to not include reheat in the analysis to simplify the analysis process, as we chose to focus more on integrating the Engineering Equation Solver (EES) code into Model Center.

Figure 1: A Typical Combined Cycle Power Plant Schematic

(Source: 1)

Model Center was used to determine the optimum design and visualize the trades-space.  A description of Model Center and the process that was followed is presented in the Trade-Off Studies section of this document.  More information on Model Center and its capabilities is available upon request.

The goal of our analysis was to maximize the production of power per unit mass flow rate of fuel.  We assumed methane as a typical fuel for our analysis.  Further analysis was performed on the three-stage heat exchanger to find the maximum reversible work in the system.
Table 1: Characteristics and Parameters Used in Analysis
	GAS TURBINE CHARACTERISTICS
	

	Compressor Exit Pressure
	1.2 MPa

	Maximum Combustor Temperature
	1200 C

	Turbine Efficiency
	90%

	Compressor Efficiency
	85%

	STEAM CYCLE CHARACTERISTICS
	

	Condenser Temperature
	40 C

	Pinch Point in Heat Exchanger
	5 C or greater

	Cycle High Pressure
	Variable

	Cycle High Temperature
	Variable

	Reheat
	ASSUMED NO REHEAT

	Feed Water Heating
	None

	Turbine Efficiency
	90%

	Feed Pump Efficiency
	60%

	AMBIENT CONDITIONS
	

	Temperature
	25 C

	Pressure
	100 kPa


Design Process Approach

The project was initially approached piecewise. The EES code for the Brayton, Rankine, and Heat Exchanger (HEX) were developed independently. Each section was checked against examples in the text for accuracy. Once accuracy was confirmed, the individual programs were integrated together to get a single EES program for the combined cycle analysis. This final “combined” code was again checked against an example in the text.

Knowing the program was functioning properly, attention was turned to completing the details of setting it up for this project. Table 1 lists the values given for this project. Also, being the goal of the analysis was to find the maximum cycle performance per unit flow rate of fuel, the mass flow rate of the Brayton cycle was fixed at 53.0998 kg/s. This value was found by working from the Lower Heating Value for Methane found in the Shapiro 5th Ed. text. This calculation can be found on page 20 of the Appendix.
The HEX was separated into three sections, as can be seen in Figure 3. The Evaporator (EV) section is where the maximum mass flow rate of the Rankine cycle can be determined. Since the Brayton flow rate is fixed, too high of a Rankine flow rate would not allow the full evaporation of water. The Economizer (EC) and Super Heater (SH) sections of HEX were defined to have equal heat flows between the Brayton and Rankine sides. An issue arose when the temperature of the exhaust of the Brayton side of the HEX was fixed. This created errors in the heat flows for the EV section of the HEX, where EES found solution convergence by allowing improper heat flow from the Rankine side (cold) to the Brayton side (hot). To remedy this, and not undermine assumed saturation states between EV and SH, it was decided that the exhaust temp of the Brayton side would be allowed to vary. This required the heat flows from the Brayton and Rankine sides to be set equal within the EV. As EES found converged solutions, it was important to make sure the exhaust temp of the Brayton side of the HEX did not drop below the temperature of the incoming Rankine flow. Care still had to be taken to ensure the pinch point was acceptable, as no fixing of this relationship was done.

The other variables allowed to vary were the Rankine cycle high pressure and high temperature variables. This means the exploration of the trade space for this project would require the varying of the Rankine high pressure, Rankine high temp, and Rankine mass flow rate. The particular variables to watch to ensure a solution was valid were the Brayton exhaust temp (to ensure it did not drop below Rankine HEX inlet temp) and the pinch point (must be greater than 5 degrees). Additionally, it was important the Rankine high temp and pressure did not exceed what is realistically possible. Research was done, and maximum values were found for both. The maximum Rankine high pressure allowed by modern steam turbines is 300 bar, or 30,000 kPa, while the maximum temperature is 873 K (2). These constraints had to be met for a realistic “ideal” solution to be found. Other assumptions made include: Air standard analysis of Brayton cycle, negligible KE and PE effects, and steady state steady flow.

Thermodynamic Model

The system schematics including the overall system and a detailed view of the heat exchanger can be seen in Figures 2 and 3.  The Brayton cycle was modeled as an open system with atmospheric temperature and pressure inlet, constant fuel consumption, and discharge to atmosphere after the heat exchanger.  The Rankine cycle is a closed-loop system with a constant temperature cold reservoir and a rather inefficient pump.  The heat exchanger is broken up into three parts: the Economizer, the Evaporator, and the Super Heater.  Each must be considered separately for proper analysis to take place. A brief discussion of the considerations in the HEX analysis can be found in the Design Process Approach.
[image: image2.emf]
Figure 2: Combined Brayton / Rankine Cycle schematic.
[image: image3.emf]
Figure 3: Detail of Heat Exchanger Breakdown.

The EES code that drives the model is included in the Appendices.  Additional code including the Model Center Wrapper code (discussed in succeeding sections) and the AutoHotkey (3) script are included in the Appendices.  The EES code was verified against a similar problem from the Shapiro 5th ed. Textbook.  The reversible work analysis EES code is included in the general EES code displayed in the Appendices.  Hand calculations of each part of the system are available upon request but are not included in this document in the interest of brevity.

Trade-Off Studies

To perform studies of the design space, we chose to use a third-party commercial design space exploration and optimization program.  We chose Model Center for ease of use, robustness, and availability.  A brief description of Model Center is presented followed by trade space and trade off studies.  This section concludes with the optimum design solution that was found.
Model Center Description

Model Center, produced by Phoenix Integration (4), is a highly sophisticated trade space exploration, visualization, and optimization package.  It is often used in aerospace and defense industry preliminary and exploratory design studies.  Major industry users include Raytheon, Boeing, Lockheed Martin, and others.  Through another course, we both were exposed to this piece of software.

Model Center operates by using “wrappers” that interface between external analysis tools such as Microsoft Excel, CATIA, StarCD, EES, and many hundreds of other stand-alone analysis programs.  In this way, Model Center is able to drive the analysis programs to explore trade spaces and find optimal solutions using a large array of programs.  Model Center plays the role of traffic cop directing variables from, for instance, a spreadsheet into a CAD program and from the CAD program to a thermal fluid program which then feeds back into a spreadsheet.  In this way, Model Center is quite sophisticated and user-friendly.  Hooking up multiple models becomes a trivial matter compared to attempting to program a link in C or equivalent programming language.
Many different exploration and optimization techniques are available within Model Center such as Monte Carlo simulators, Gradient Function optimizers, Darwin Evolution models, Genetic Algorithms, and others.  A full suite of statistical options are available under each regime.  
The Design Space Explorer provides an interface to a suite of software visualization tools otherwise known as ATSV that is developed by Penn State.  3-D Glyph Plots, Surface Contour Plots, Parallel Coordinate Plots, and many other types of useful graphics are available in the package.  With the appropriate 3-D goggles, many of the graphs become quite eye-popping.
Model Center has support for EES analysis.  However, because OSU has chosen to purchase the “Commercial” version of EES rather than the “Professional” version, these abilities are very crippled.  The main difference, as far as we can tell, between Commercial and Professional is the inability to run EES from the command line.

A workaround to this problem was found by using a freeware program called AutoHotkey that creates keyboard and mouse command macros.  Calling a macro custom-written to interface EES with Model Center allowed a somewhat dysfunctional union to once again be established.  However, because we had to use a hack and work around to the problem of buying the cheaper version of EES, we were greatly hampered in fully exploring the design space.  What should have taken an hour or less took over twelve to generate enough points for proper analysis.  In the future, it would be highly useful to have the Professional version of EES available to those who wish to use it.

Results of Trade-Space Exploration

To properly explore the trade space, initial analysis was run using a Monte Carlo simulator set to distribute points throughout the proposed design space.  Many points failed to validate in EES.  The ones that did began to form a general pattern.  Further analysis was performed on the most promising points using Darwin Genetic Algorithms and Gradient Analysis.  The resulting set of data points is displayed in Figure 4.  The points in color are considered feasible while the points that are gray are not feasible due to constraint violations.

Figure 4: Display of all datapoints generated in Model Center shown including infeasible points (in gray).  Cycle High Pressure versus Cycle High Temperature versus Mass Flow Rate of Rankine Cycle are graphed on their respective axes.
Figure 5 displays the points that are feasible in the design space.  Note that there appears to be shape and structure to the feasible points.  In fact, they come down to a rather pointy end leading to hope of a single optimum solution.

Figure 5: Display feasible points only.  Cycle High Pressure versus Cycle High Temperature versus Mass Flow Rate of Rankine Cycle are graphed on their respective axes.

Figure 6 provides a representative screenshot of the constraint definitions that we employed in Model Center.  The most critical constraints were: maintaining a positive pinch point value (greater than 5 deg), maintaining the Brayton cycle exhaust temperature above the Rankine cycle inlet temperature, not exceeding the maximum standard turbine temperature operating conditions (as discussed in the Design Process Approach section), and not exceeding the maximum standard turbine pressure operating conditions.  Several other areas, such as negative absolute temperature values, were also controlled both in Model Center and EES.

Figure 7 is a representative screenshot of the objectives that Model Center used to determine the Pareto Frontier.  We chose to maximize the overall combined cycle efficiency and the overall net work while minimizing the turbine pressure and temperature (as this would hope to aid in cost savings).  The Pareto Frontier displayed in Figure 8 resulted from this definition.  If other objectives were used, a different Pareto Frontier would have resulted.  It should be noted that our final point was chosen independently of turbine pressure and temperature although a reality check was performed to throw out pressures and temperatures that are beyond standard turbine operating conditions.

Figure 6: Representative screenshot of the constraint definitions used in Model Center.

Figure 7: Representative screenshot of the Objective Definitions in Model Center used to find the Pareto Frontier.

Figure 8: Visualization of Pareto Frontier points.  Cycle High Pressure versus Cycle High Temperature versus Mass Flow Rate of Rankine Cycle are graphed on their respective axes.

To gain an understanding of the design space surfaces that we are operating on, the surface contour plots in Figure 9 were created.  They show the Total Net Work per Mass Flow Rate with respect to Cycle High Pressure, Cycle High Temperature, and Mass Flow Rate of the Rankine Cycle.  Note the two peaks on the bottom left graph.  Also note that there are peaks on the other two graphs.  These correspond to the optimized design point.  Had the design space not been explored, a normal optimization algorithm would have fixated on one or the other of the maxima present in the third graph.  This could have resulted in a local maxima being found to the exclusion of the global maxima.  Figure 10 provides another view of the data.  In this graph, it is easy to see that turbine pressure does not significantly affect the total net work output.

An understanding of which variables actually vary and in which direction the best variation occurs can be garnered from Figure 11.  This graph clearly shows that several variables do not vary.  Some of these variables are actually fixed in the EES code while others do not vary for other reasons related to the system analysis.  It is very informative to examine where each of the variables finds optimum solutions.
[image: image9.jpg]Data Explorer (Data Import PlugIn):

Data Edt Page

L BB P e |l stondadrioss - [ & |pata visuslzer v

[ ey

Drag Let huttonto Zor
EES.

51700 1775
5172 5172
3] P

2 - DV Star Glyph

‘.

3 - Variable Influence

A5 Cyet igh temp

Cyele_pign pres

[4-Variable Influence]

5 - DV Parallel Coordinate:

il

6 - Prediction Profiler

‘ >

For Help, press F1 [ [

) start| =) windows Task Manager [ 4 Phoenix Integration .. i Design viewer -Design 2... | £ images | B3 dobe Photoshop cs3E... | 20 SER 1zzm





Figure 9: Surface contour plots of Total Net Work per Mass Flow Rate with respect to Cycle High Pressure, Cycle High Temperature, and Mass Flow Rate of Rankine Cycle.  

Figure 10: Graphs of Total Net Work per Mass Flow Rate versus Cycle High Pressure, Cycle High Temperature, and Mass Flow Rate of Rankine Cycle.  Note that pressure does not seem to affect the total net work significantly.

Figure 11: Parallel Coordinate plot of all design variables.
Selected Design Overview

From the analysis performed in Model Center, an optimum design point was found.  On run 2524 out of approximately 3500 runs, the point shown in the Appendices was found.  This relates to the lowest and most forward point in Figure 5 as shown previously.  The full dataset, viewable in the Appendices, shows that all performance criteria were met.  The cycle is within temperature and pressure specs for the heat exchanger and the steam turbine.  Total net work is 29770 kW.  Total cycle efficiency is 0.5954, which is a fairly accurate number when compared to highly efficient real-world systems that can very closely approach 0.6 efficiency (2).
Given the optimal design, the mass flow rates of the Brayton and Rankine cycle were scaled back to give a Brayton mass flow rate of unity. The values found from this solution were put into Table 2, as requested, so the evaluation of the HEX could be easily carried out.

Table 2: Heat Exchanger Table
	Quantity
	Overall HEX
	Economizer
	Evaporator
	Super Heater

	Duty (Q exchange) [kW]
	567.8
	340.7
	60.44
	166.7

	Brayton m_dot in [kg/s]
	1
	-------
	-------
	-------

	Brayton m_dot out [kg/s]
	1
	-------
	-------
	-------

	Rankine m_dot in [kg/s]
	0.1949
	-------
	-------
	-------

	Rankine m_dot out [kg/s]
	0.1949
	-------
	-------
	-------

	Brayton T in [K]
	865.3
	656.7
	713.2
	865.3

	Brayton T out [K]
	326
	326
	656.7
	713.2

	Rankine T in [K]
	317.3
	317.3
	645.8
	645.8

	Rankine T out [K]
	745.9
	645.8
	645.8
	745.9


Reversible Work Analysis

Since the authors are both in the graduate student section, an additional analysis was done on the reversible work in the HEX. The results can be found in Table 3 given below. The hand written equations derived can be seen in the Appendix on page 21.
Table 3: Reversible Work Analysis at Optimum Design Point

	Quantity
	Overall HEX
	Economizer
	Evaporator
	Super Heater

	Rev Work Brayton [kW]
	15530
	6717
	1811
	7002

	Rev Work Rankine [kW]
	-13555
	-6891
	-3210
	-3455

	Rev Work Total [kW]
	1974
	-174.2
	-1398
	3547


References
1. http://www.cogeneration.net/Combined_Cycle_Power_Plants.htm
2. http://www.powergeneration.siemens.com/products-solutions-services/products-packages/steam-turbines/scale+larger+150mw/sst-6000-series/sst-6000-series.htm
3. http://www.autohotkey.com/
4. http://www.phoenix-int.com/
Appendix: Code and Related Errata

Model Center EES Modified Wrapper
#

# @description Spring analysis using Engineering Equation Solver

# @version DEMO

# @author Phoenix Integration

#

RunCommands

{

   generate inFile

   # start up the ees.exe program using a macro file

   # this line may need to be modified to specify the

   # full path for ees.exe

   #run "c:\\ees_av\\ees.exe S:\\Gibson\\ME544\\combined cycle analysis.ees"

   run s:\\gibson\\me544\\ees.ahk

   parse outFile

}

RowFieldInputFile inFile

{

   templateFile: S:\Gibson\ME544\combined.in.csv.template

   fileToGenerate: S:\Gibson\ME544\combined.in.csv

   setDelimiters ","

   variable: Cycle_high_pres  double 1 1 description="Cycle High Pressure" units="kPa"

   variable: Cycle_high_temp double 1 2 description="Cycle High Temperature" units="K"

   variable: m_dot_r double 1 3 description="Mass Flow Rate Brayton Cycle" units="kg/s"

   }

RowFieldOutputFile outFile

{

   fileToParse: S:\Gibson\ME544\combined.out.csv

   setDelimiters ","

   variable: W_dot_net_tot double 1 1 description="Net work of combined cycle" units="kW"

   variable: W_dot_net_b double 1 2 description="Net work of Brayton cycle" units="kW"

   variable: W_dot_c_b double 1 3 description="work consumed by Brayton compressor" units="kW"

   variable: W_dot_t_b double 1 4 description="work produced by Brayton turbine" units="kW"

   variable: W_dot_net_r double 1 5 description="Net work of Rankine cycle" units="kW"

   variable: W_dot_p_r double 1 6 description="work consumed by Rankine pump" units="kW"

   variable: W_dot_t_r double 1 7 description="work produced by Rankine turbine" units="kW"

   variable: Q_dot_in_b double 1 8 description="heat transfer into Brayton side" units="kW"

   variable: Q_dot_out_b double 1 9 description="heat transfer out of Brayton side" units="kW"

   variable: Q_dot_in_r double 1 10 description="heat transfer into Rankine side" units="kW"

   variable: Q_dot_out_r double 1 11 description="heat transfer out of Rankine side" units="kW"

   variable: n_cyc_tot double 1 12 description="Thermal eff of combined cycle"

   variable: n_cyc_b double 1 13 description="Thermal eff of Brayton cycle"

   variable: n_cyc_r double 1 14 description="Thermal eff of Ranikine cycle"

   variable: bwr_b double 1 15 description="back work ratio of Brayton cycle"

   variable: bwr_r double 1 16 description="back work ratio of Rankine cycle"

   variable: Pinch_point double 1 17 description="pinch point in HEX" units="K"

   variable: Actual_flow_ratio double 1 18 description="actual ratio of m_dot_r/m_dot_b"

   variable: Needed_flow_ratio double 1 19 description="ratio of m_dot_r/m_dot_b needed to get evaporation in Rankine side"

   variable: Q_dot_EC_r double 1 20 description="duty of economizer" units="kW"

   variable: Q_dot_EV_r double 1 21 description="duty of evaporator" units="kW"

   variable: Q_dot_SH_r double 1 22 description="duty of super heater" units="kW"

   variable: Q_dot_HEX_r double 1 23 description="duty of HEX overall" units="kW"

   variable: T4_b double 1 24 description="Brayton super heater inlet temp" units="K"

   variable: T_bs double 1 25 description="Brayton superheater outlet/evaporator inlet temp" units="K"

   variable: T_bv double 1 26 description="Brayton evaporator outlet/economizer inlet temp" units="K"

   variable: T5_b double 1 27 description="Brayton economizer temp (final exhaust temp)" units="K"

   variable: T4_r double 1 28 description="Rankine economizer inlet temp" units="K"

   variable: T_rv double 1 29 description="Rankine economizer outlet/evaporator inlet temp" units="K"

   variable: T_rs double 1 30 description="Rankine evaporator outlet/super heater inlet temp" units="K"

   variable: T1_r double 1 31 description="Rankine super heater outlet temp (Cycle_high_temp)" units="K"

}

AutoHotkey Script

Run c:\ees_av\EES.exe S:\Gibson\ME544\Combined Cycle 04.EES

sleep 1000

Send c

sleep 1500

Send {F2}

sleep 3000

WinKill EES Academic Commercial, , 1

sleep 2000
EES Combined Cycle Code

"read in input values from a file"
$import 'combined.in.csv' Cycle_high_pres, Cycle_high_temp, m_dot_r
///-----**** COMBINED CYCLE ANANLYSIS ****-----///
"Inputs to system"

"Ambient Conditions"

P_amb=100 [kPa]


T_amb=298.15 [K]


"Brayton Characteristics"

Exhaust_temp=T5_b "....let this vary, but pay attention it doesn't get below cold side inlet temp."

Compressor_exit_pres=1200 [kPa]


Combustor_max_temp=1473.15 [K]


Turbine_eff_b=0.9


Comp_eff_b=0.85


m_dot_b=53.0998 [kg/s]


"Rankine Characteristics"

Condenser_temp=313.15 [K]


Turbine_eff_r=0.90


Pump_eff_r=0.60


"Rankine variables to vary"

"Cycle_high_pres=8000 [kPa]


Cycle_high_temp=673.15 [K]


m_dot_r=9 [kg/s]"

"Methane fuel heating values"

methane_LHV=50020 [kJ/kg]


methane_HHV=55510 [kJ/kg]

//Brayton Cycle - Air-Standard Analysis//

"compresor inlet conditions"

P1_b=P_amb


T1_b=T_amb


"m_dot_b=1 [kg/s]"

"compression ratio"

"C_ratio=10"

"P2=C_ratio*P1"

P2_b=Compressor_exit_pres


"efficiencies"

n_c_b=Comp_eff_b


n_t_b=Turbine_eff_b


n_reg_b=0.00


"compressor"

W_dot_c_b=m_dot_b*(h2_b-h1_b)


s1_b=s2_b


n_c_b=(h2s_b-h1_b)/(h2_b-h1_b)


"regenerator"

n_reg_b=(hx_b-h2_b)/(h4_b-h2_b)


Q_dot_in_b=m_dot_b*(h3_b-hx_b)


m_dot_fuel_LHV=Q_dot_in_b/methane_LHV


m_dot_fuel_HHV=Q_dot_in_b/methane_HHV


"turbine"

T3_b=Combustor_max_temp


W_dot_t_b=m_dot_b*(h3_b-h4_b)


s3_b=s4_b


n_t_b=(h3_b-h4_b)/(h3_b-h4s_b)


"heat transfer"

"Q_in=h3-h2...comment out due to regenerator"

Q_dot_out_b=m_dot_b*(h4_b-h5_b)


"cycle values"

n_cyc_b=((W_dot_t_b/m_dot_b)-(W_dot_c_b/m_dot_b))/(Q_dot_in_b/m_dot_b)


bwr_b=(W_dot_c_b/m_dot_b)/(W_dot_t_b/m_dot_b)


W_dot_net_b=W_dot_t_b-W_dot_c_b


"1-2 analysis"

h1_b=Enthalpy(Air,T=T1_b)


s1_b=Entropy(Air,T=T1_b,P=P1_b)


s2_b=Entropy(Air,T=T2s_b,P=P2_b)


h2s_b=Enthalpy(Air,T=T2s_b)


h2_b=Enthalpy(Air,T=T2_b)


"2-3 analysis"

P2_b=P3_b


"3-4 analysis"

h3_b=Enthalpy(Air,T=T3_b)


s3_b=Entropy(Air,T=T3_b,P=P3_b)


s4_b=Entropy(Air,T=T4s_b,P=P4_b)


h4s_b=Enthalpy(Air,T=T4s_b)


h4_b=Enthalpy(Air,T=T4_b)


"4-5 analysis"

P4_b=P1_b


"T5_b=Exhaust_temp  ........commented out because overconstrained the HEX"

h5_b=Enthalpy(Air,T=T5_b)

//Rankine Power Cycle - Real Substance Analysis (Steam)//

"turbine inlet conditions"

P1_r=Cycle_high_pres


T1_r=Cycle_high_temp


"x1=1


T1_r=T_sat(Steam,P=P1_r)......comment out due to superheater in HEX"

"m_dot_r=((4.449e5)/3600)*1[kg/s]"

"efficiencies"

n_p_r=Pump_eff_r


n_t_r=Turbine_eff_r


"pump"

W_dot_p_r=m_dot_r*(h4_r-h3_r)


s3_r=s4_r


n_p_r=(h4s_r-h3_r)/(h4_r-h3_r)


"turbine"

W_dot_t_r=m_dot_r*(h1_r-h2_r)


s1_r=s2_r


n_t_r=(h1_r-h2_r)/(h1_r-h2s_r)


"heat transfer"

Q_dot_in_r=m_dot_r*(h1_r-h4_r)


Q_dot_out_r=m_dot_r*(h2_r-h3_r)


"cycle values"

n_cyc_r=((W_dot_t_r/m_dot_r)-(W_dot_p_r/m_dot_r))/(Q_dot_in_r/m_dot_r)


bwr_r=(W_dot_p_r/m_dot_r)/(W_dot_t_r/m_dot_r)


W_dot_net_r=W_dot_t_r-W_dot_p_r


"1-2 analysis"

T2_r=T3_r


"h1_r=Enthalpy(Steam,P=P1_r,x=x1)....comment out due to superheater in HEX"

h1_r=Enthalpy(Steam,P=P1_r,T=T1_r)


"s1_r=Entropy(Steam,P=P1_r,x=x1).....comment out due to superheater in HEX"

s1_r=Entropy(Steam,P=P1_r,T=T1_r)


s2_r=Entropy(Steam,T=T2_r,x=x2s)


h2s_r=Enthalpy(Steam,T=T2_r,x=x2s)


h2_r=Enthalpy(Steam,T=T2_r,x=x2)


P2_r=P_sat(Steam,T=T2_r)


"2-3 analysis"

T3_r=Condenser_temp


x3=0


h3_r=Enthalpy(Steam,T=T3_r,x=x3)


s3_r=Entropy(Steam,T=T3_r,x=x3)


P3_r=P_sat(Steam,T=T3_r)


"3-4 analysis"

s4_r=Entropy(Steam,T=T4s_r,P=P4_r)


h4s_r=Enthalpy(Steam,T=T4s_r,P=P4_r)


h4_r=Enthalpy(Steam,T=T4_r,P=P4_r)


"4-1 analysis"

P4_r=P1_r

//Heat Echanger for combined cycle//

"Brayton Side"

P_bv=P1_b


h_bv=Enthalpy(Air,T=T_bv)


h_bs=Enthalpy(Air,T=T_bs)


"Rankine Side"

P_rv=P4_r


x_v=0


h_rv=Enthalpy(Steam,P=P_rv,x=x_v)


T_rv=T_sat(Steam,P=P_rv)


P_rs=P4_r


x_s=1


h_rs=Enthalpy(Steam,P=P_rs,x=x_s)


T_rs=T_sat(Steam,P=P_rs)


"Economizer"

m_dot_r*(h_rv-h4_r)=-m_dot_b*(h5_b-h_bv)


Q_dot_EC_r=m_dot_r*(h_rv-h4_r)


Q_dot_EC_b=m_dot_b*(h5_b-h_bv)


"Evaporator"

Actual_flow_ratio=m_dot_r/m_dot_b


Needed_flow_ratio=-(h_bv-h_bs)/(h_rs-h_rv)


Pinch_point=T_bv-T_rv


Q_dot_EV_r=m_dot_r*(h_rs-h_rv)


Q_dot_EV_b=m_dot_b*(h_bv-h_bs)


Q_dot_EV_r=-Q_dot_EV_b "......commented out because it set the Actual and Needed flows always equal"

"Super Heater"

m_dot_r*(h1_r-h_rs)=-m_dot_b*(h_bs-h4_b)


Q_dot_SH_r=m_dot_r*(h1_r-h_rs)


Q_dot_SH_b=m_dot_b*(h_bs-h4_b)


"Overall HEX"

Q_dot_HEX_r=Q_dot_EC_r+Q_dot_EV_r+Q_dot_SH_r


Q_dot_HEX_b=Q_dot_EC_b+Q_dot_EV_b+Q_dot_SH_b

//Combined Cycle Calculations//

W_dot_net_tot=W_dot_net_b+W_dot_net_r


n_cyc_tot=(W_dot_net_b+W_dot_net_r)/(Q_dot_in_b)

//Exergy Analysis of HEX//


"Brayton Entropy values"

s_bs=Entropy(Air,T=T_bs,P=P4_b)


s_bv=Entropy(Air,T=T_bv,P=P4_b)


s5_b=Entropy(Air,T=T5_b,P=P4_b)


"Rankine Entropy values"


s_rv=Entropy(Steam,T=T_rv,P=P_rv)


s_rs=Entropy(Steam,T=T_rs,P=P_rs)


"Economizer"


W_rev_EC_b=m_dot_b*((h_bv-h5_b)-T_amb*(s_bv-s5_b))


W_rev_EC_r=m_dot_r*((h4_r-h_rv)-T_amb*(s4_r-s_rv))


W_rev_EC_tot=W_rev_EC_b+W_rev_EC_r


"Evaporator"


W_rev_EV_b=m_dot_b*((h_bs-h_bv)-T_amb*(s_bs-s_bv))


W_rev_EV_r=m_dot_r*((h_rv-h_rs)-T_amb*(s_rv-s_rs))


W_rev_EV_tot=W_rev_EV_b+W_rev_EV_r


"Super Heater"


W_rev_SH_b=m_dot_b*((h4_b-h_bs)-T_amb*(s4_b-s_bs))


W_rev_SH_r=m_dot_r*((h_rs-h1_r)-T_amb*(s_rs-s1_r))


W_rev_SH_tot=W_rev_SH_b+W_rev_SH_r


"Overall HEX"


W_rev_HEX_b=m_dot_b*((h4_b-h5_b)-T_amb*(s4_b-s5_b))


W_rev_HEX_r=m_dot_r*((h4_r-h1_r)-T_amb*(s4_r-s1_r))


W_rev_HEX_tot=W_rev_HEX_b+W_rev_HEX_r
"write out output values to a file"
$export 'combined.out.csv' W_dot_net_tot, W_dot_net_b, W_dot_c_b, W_dot_t_b, W_dot_net_r, W_dot_p_r, W_dot_t_r, Q_dot_in_b, Q_dot_out_b, Q_dot_in_r, Q_dot_out_r, n_cyc_tot, n_cyc_b, n_cyc_r, bwr_b, bwr_r, Pinch_point, Actual_flow_ratio, Needed_flow_ratio, Q_dot_EC_r, Q_dot_EV_r, Q_dot_SH_r, Q_dot_HEX_r, T4_b, T_bs, T_bv, T5_b, T4_r, T_rv, T_rs, T1_r
Appendix: Hand Calculations and Diagrams

Appendix: Optimum Solution Data

Optimal Solution From Model Center:
	design variable (Model.combined_cycle_EES.Cycle_high_pres)
	21701

	design variable (Model.combined_cycle_EES.Cycle_high_temp)
	745.9

	design variable (Model.combined_cycle_EES.m_dot_r)
	10.351

	response (Model.combined_cycle_EES.Actual_flow_ratio)
	0.194935

	response (Model.combined_cycle_EES.bwr_b)
	0.512469

	response (Model.combined_cycle_EES.bwr_r)
	0.031857

	response (Model.combined_cycle_EES.n_cyc_b)
	0.367208

	response (Model.combined_cycle_EES.n_cyc_r)
	0.378458

	response (Model.combined_cycle_EES.n_cyc_tot)
	0.595431

	response (Model.combined_cycle_EES.Needed_flow_ratio)
	0.194935

	response (Model.combined_cycle_EES.Pinch_point)
	10.97051

	response (Model.combined_cycle_EES.Q_dot_EC_r)
	18091.89

	response (Model.combined_cycle_EES.Q_dot_EV_r)
	3209.617

	response (Model.combined_cycle_EES.Q_dot_HEX_r)
	30151.03

	response (Model.combined_cycle_EES.Q_dot_in_b)
	49998.95

	response (Model.combined_cycle_EES.Q_dot_in_r)
	30151.03

	response (Model.combined_cycle_EES.Q_dot_out_b)
	30151.03

	response (Model.combined_cycle_EES.Q_dot_out_r)
	18740.13

	response (Model.combined_cycle_EES.Q_dot_SH_r)
	8849.522

	response (Model.combined_cycle_EES.T_bs)
	713.1719

	response (Model.combined_cycle_EES.T_bv)
	656.7363

	response (Model.combined_cycle_EES.T_rs)
	645.7657

	response (Model.combined_cycle_EES.T_rv)
	645.7657

	response (Model.combined_cycle_EES.T1_r)
	745.9

	response (Model.combined_cycle_EES.T4_b)
	865.2598

	response (Model.combined_cycle_EES.T4_r)
	317.3018

	response (Model.combined_cycle_EES.T5_b)
	326.0213

	response (Model.combined_cycle_EES.W_dot_c_b)
	19299.19

	response (Model.combined_cycle_EES.W_dot_net_b)
	18360.02

	response (Model.combined_cycle_EES.W_dot_net_r)
	11410.9

	response (Model.combined_cycle_EES.W_dot_net_tot)
	29770.92

	response (Model.combined_cycle_EES.W_dot_p_r)
	375.4816

	response (Model.combined_cycle_EES.W_dot_t_b)
	37659.21

	response (Model.combined_cycle_EES.W_dot_t_r)
	11786.38


Optimal Solution Results From EES:

Actual_flow_ratio=0.1949 

bwr_b=0.5125 

bwr_r=0.03186 

Combustor_max_temp=1473 [K]

Compressor_exit_pres=1200 [kPa]

Comp_eff_b=0.85 

Condenser_temp=313.2 [K]

Cycle_high_pres=21701 [kPa]

Cycle_high_temp=745.9 [K]

Exhaust_temp=326 [K]

h1_b=298.6 [kJ/kg]

h1_r=3117 [kJ/kg]

h2s_b=607.5 [kJ/kg]

h2s_r=1851 [kJ/kg]

h2_b=662 [kJ/kg]

h2_r=1978 [kJ/kg]

h3_b=1604 [kJ/kg]

h3_r=167.5 [kJ/kg]

h4s_b=815.6 [kJ/kg]

h4s_r=189.3 [kJ/kg]

h4_b=894.4 [kJ/kg]

h4_r=203.8 [kJ/kg]

h5_b=326.6 [kJ/kg]

hx_b=662 [kJ/kg]

h_bs=727.8 [kJ/kg]

h_bv=667.3 [kJ/kg]

h_rs=2262 [kJ/kg]

h_rv=1952 [kJ/kg]

methane_HHV=55510 [kJ/kg]

methane_LHV=50020 [kJ/kg]

m_dot_b=53.1 [kg/s]

m_dot_fuel_HHV=0.9007 [kg/s]

m_dot_fuel_LHV=0.9996 [kg/s]

m_dot_r=10.35 [kg/s]

Needed_flow_ratio=0.1949 

n_cyc_b=0.3672 

n_cyc_r=0.3785 

n_cyc_tot=0.5954 

n_c_b=0.85 

n_p_r=0.6 

n_reg_b=0 

n_t_b=0.9 

n_t_r=0.9 

P1_b=100 [kPa]

P1_r=21701 [kPa]

P2_b=1200 [kPa]

P2_r=7.381 [kPa]

P3_b=1200 [kPa]

P3_r=7.381 [kPa]

P4_b=100 [kPa]

P4_r=21701 [kPa]

Pinch_point=10.97 [K]

Pump_eff_r=0.6 

P_amb=100 [kPa]

P_bv=100 [kPa]

P_rs=21701 [kPa]

P_rv=21701 [kPa]

Q_dot_EC_b=-18092 [kW]

Q_dot_EC_r=18092 [kW]

Q_dot_EV_b=-3210 [kW]

Q_dot_EV_r=3210 [kW]

Q_dot_HEX_b=-30151 [kW]

Q_dot_HEX_r=30151 [kW]

Q_dot_in_b=49999 [kW]

Q_dot_in_r=30151 [kW]

Q_dot_out_b=30151 [kW]

Q_dot_out_r=18740 [kW]

Q_dot_SH_b=-8850 [kW]

Q_dot_SH_r=8850 [kW]

s1_b=5.699 [kJ/kg·K]

s1_r=5.95 [kJ/kg·K]

s2_b=5.699 [kJ/kg·K]

s2_r=5.95 [kJ/kg·K]

s3_b=6.713 [kJ/kg·K]

s3_r=0.5723 [kJ/kg·K]

s4_b=6.713 [kJ/kg·K]

s4_r=0.5723 [kJ/kg·K]

s5_b=5.789 [kJ/kg·K]

s_bs=6.596 [kJ/kg·K]

s_bv=6.508 [kJ/kg·K]

s_rs=4.202 [kJ/kg·K]

s_rv=4.202 [kJ/kg·K]

T1_b=298.2 [K]

T1_r=745.9 [K]

T2s_b=600.2 [K]

T2_b=651.8 [K]

T2_r=313.2 [K]

T3_b=1473 [K]

T3_r=313.2 [K]

T4s_b=794 [K]

T4s_r=313.8 [K]

T4_b=865.3 [K]

T4_r=317.3 [K]

T5_b=326 [K]

Turbine_eff_b=0.9 

Turbine_eff_r=0.9 

T_amb=298.2 [K]

T_bs=713.2 [K]

T_bv=656.7 [K]

T_rs=645.8 [K]

T_rv=645.8 [K]

W_dot_c_b=19299 [kW]

W_dot_net_b=18360 [kW]

W_dot_net_r=11411 [kW]

W_dot_net_tot=29771 [kW]

W_dot_p_r=375.5 [kW]

W_dot_t_b=37659 [kW]

W_dot_t_r=11786 [kW]

W_rev_EC_b=6717 [kW]

W_rev_EC_r=-6891 [kW]

W_rev_EC_tot=-174.2 [kW]

W_rev_EV_b=1811 [kW]

W_rev_EV_r=-3210 [kW]

W_rev_EV_tot=-1398 [kW]

W_rev_HEX_b=15530 [kW]

W_rev_HEX_r=-13555 [kW]

W_rev_HEX_tot=1974 [kW]

W_rev_SH_b=7002 [kW]

W_rev_SH_r=-3455 [kW]

W_rev_SH_tot=3547 [kW]

x2=0.7525 

x2s=0.6999 

x3=0 

x_s=1 

x_v=0

